Figure 9-9 *(continued)*

```
; The TX subroutine sends out the byte passed to it in W.
; It returns with Z=1 if ACK occurs.
; It returns with Z=0 if NOACK occurs

TX
        movwf   TXBUFF          ;Save parameter in TXBUFF
        bsf     STATUS,C        ;Rotate a one through TXBUFF to count bits
TX_1
        rlf     TXBUFF,F        ;Rotate TXBUFF left, through Carry
        movf    TXBUFF,F        ;Set Z bit when all eight bits have been transferred
        btfss   STATUS,Z        ;Until Z=1
        call    BitOut          ; send Carry bit, then clear Carry bit
        btfss   STATUS,Z        ;
        goto    TX_1            ;   then do it again
        call    BitIn           ;Read acknowledge bit into bit 0 of RXBUFF
        movlw   B'0000001'      ;Check acknowledge bit
        andwf   RXBUFF,W        ;Z=1 if ACK; Z=0 if NOACK
        return

; The RX subroutine receives a byte from the I2C bus into W, using RXBUFF buffer
; Call RX with bit 7 of TXBUFF clear for ACK.
; Call RX with bit 7 of TXBUFF set for NOACK.

RX
        movlw   B'00000001      ;Rotate a one through RXBUFF to the carry bit to count bits
        movwf   RXBUFF
RX_1
        rlf     RXBUFF,F        ;Shift previous bits left
        call    BitIn           ;Read a bit from SDA into bit 0 of RXBUFF
        btfss   STATUS,C        ;C=1 yet;
        goto    RX_1            ;No, do it again
        rlf     TXBUFF,F        ;Move bit 7 of TXBUFF to Carry bit
        call    BitOut          ;and from there to SDA as acknowledgment
        movf    RXBUFF,W        ;Put received byte into W
        return

; The BitOut subroutine transmits, then clears, the Carry bit

BitOut
        bcf     INDF,SDA        ;Copy Carry bit to SDA
        btfsc   STATUS,C
        bsf     INDF,SDA
        bsf     INDF,SCL        ;Pulse clock line
        delay   0,1,2           ;t:HIGH
        bcf     INDF,SCL
        bcf     STATUS,C        ;Clear Carry bit
        return

; The BitIn subroutine receives one bit into bit 0 of RXBUFF

BitIn
        bsf     INDF,SDA        ;Release SDA line
        bsf     INDF,SCL        ;Drive clock line high
        bcf     RXBUFF,0        ;Copy SDA to bit 0 of RXBUFF
        btfsc   PORTC,SDA
        bsf     RXBUFF,0
        bcf     INDF,SCL        ;Drive clock line low again
        return

;;;;;;; End of I2C subroutines ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
```

9.5 TEMPERATURE SENSOR

The combination of an analog temperature transducer, an analog-to-digital converter, and an I²C bus interface all in a tiny SO-8 surface-mount package represents a significant contribution to designers.



Figure 9-10 DAC o

The analog voltage from
physical proximity that
inside the chip, once and

National Semicondu
with ±2°C accuracy. Th
to +125°C. For many
obtained with the suppo
output. This 0.5°C resolu